
#-------------------------------------------------
#
# Project created by QtCreator 2011-07-07T08:39:24
#
#-------------------------------------------------

QT       += core gui multimedia network

TARGET = wspr
VERSION = 0.1.0
TEMPLATE = app

win32 {
F90 = g95
g95.output = ${QMAKE_FILE_BASE}.o
g95.commands = $$F90 -c -O2 -o ${QMAKE_FILE_OUT} ${QMAKE_FILE_NAME}
g95.input = F90_SOURCES
QMAKE_EXTRA_COMPILERS += g95
}

unix {
F90 = gfortran
gfortran.output = ${QMAKE_FILE_BASE}.o
gfortran.commands = $$F90 -c -O2 -o ${QMAKE_FILE_OUT} ${QMAKE_FILE_NAME}
gfortran.input = F90_SOURCES
QMAKE_EXTRA_COMPILERS += gfortran
}

SOURCES += main.cpp mainwindow.cpp plotter.cpp about.cpp \
           soundin.cpp soundout.cpp barmeter.cpp devsetup.cpp \
    widegraph.cpp \
    decodethread.cpp

F90_SOURCES = symspec.f90 four2a.f90 recvpkt.f90

HEADERS  += mainwindow.h plotter.h soundin.h soundout.h \
            palette_linrad.h about.h barmeter.h \
    devsetup.h \
    widegraph.h \
    decodethread.h

DEFINES += __cplusplus

FORMS    += mainwindow.ui about.ui devsetup.ui widegraph.ui

RC_FILE = map65.rc

unix {
INCLUDEPATH += $$quote(/usr/include/QtMultimediaKit)
LIBS += $$quote(-lQtMultimediaKit) -lfftw3f /usr/lib/libgfortran.so.3
}

win32 {
LIBS += ../trunk/libfftw3f_win.a
#LIBS += ../trunk/lib/libastro.a
LIBS += libwsock32
LIBS += C:/MinGW/lib/libf95.a
}
